COMPASS for Pressure (or Flow)
Alert if ambient conditions (pressure, temperature, relative humidity) are outside limits

Create a test macro to set ambient limits and specify what to do if conditions are outside the limits

'**************************************************
'Test Macros do not have a return value.
'Manipulate the test or device collection as desired. 
'iT  The current temperature point in the test
'iL  The current line pressure point in the test.
'iC  The current pressure cycle in the test.
'IP  The current pressure point in the test.
'cTest   The test class .
'cConfig  Configuration class that holds all active devices.
'
'**************************************************
Function AmbLimitMonitor(iT, iL, iC, iP, cTest, cConfig)

'enter e-mail address if using the e-mail notification feature
'The method used relies on MS Outlook(R) being open when running the test
'comment out the email lines if not using this feature
emailto = "noreply@flukecal.com"

[bookmark: _GoBack]Delay = 900 '(sec), time to wait between email notifications

'enter limits and units as desired
PrsAmbLL = 85
PrsAmbHL = 115
PrsAmbLtxt = "kPa"	'Set this to mbar, kPa, psi or  hPa
PrsAmbLunit = 3		'Set to corresponding integer: Pa=1, mbar=2, kPa=3
					'mmHG0C=6, Torr=7, psi=9, inHg32F=26, hPa=38
'	cDebug.LogStatus "Amb Pressure Low Limit: " & PrsAmbLL & " " & PrsAmbLtxt & " (" & PrsAmbLunit & ")"

TempAmbLL = 15
TempAmbHL = 25
TempAmbLtxt = "C"	'Set to C or F
TempAmbLunit = 0	'Set to corresponding integer; C=0, F=1, K=2, R=3

RH_AmbLL = 20 '%
RH_AmbHL =  70 '%
RH_AmbLtxt = "%"

PrsAmb = cCompass.cConfig.AmbPrs.FinalOutput
PrsAmbU = cCompass.cConfig.AmbPrs.RangeMain.UnitFinal
PrsAmbUtxt = cCompass.cConfig.AmbPrs.RangeMain.UnitFinalText
'	cDebug.LogStatus "AmbPrs = " & PrsAmb & " " & PrsAmbUtxt & " (" & PrsAmbU & ")"
PrsLL = cCOMPASS.UnitConversion (CDbl(PrsAmbLL),CInt(PrsAmbU),CInt(PrsAmbLunit),0)
'								(input,newUnit,curUnit,outputType: 0-pressure, 1-temperature, 10-Humidity
'	cDebug.LogStatus "AmbPrsLL = " & PrsLL & " " & PrsAmbUtxt & " (" & PrsAmbU & ")"
PrsHL = cCOMPASS.UnitConversion (CDbl(PrsAmbHL),CInt(PrsAmbU),CInt(PrsAmbLunit),0)
'	cDebug.LogStatus "AmbPrsHL = " & PrsHL & " " & PrsAmbUtxt & " (" & PrsAmbU & ")"
If PrsAmb < PrsLL  Then
	msg = "Amb Pressure is too low: " & PrsAmb & " " & PrsAmbUtxt
	ambAlert msg,emailto,delay
ElseIf PrsAmb > PrsHL Then 
	msg = "Amb Pressure is too high: " & PrsAmb & " " & PrsAmbUtxt
	ambAlert msg,emailto,delay
End If

TempAmb = cCompass.cConfig.AmbTemp.FinalOutput
TempAmbU = cCompass.cConfig.AmbTemp.RangeMain.UnitFinal
TempAmbUtxt = cCompass.cConfig.AmbTemp.RangeMain.UnitFinalText
'	cDebug.LogStatus "AmbTemp " & TempAmb & " " & TempAmbUtxt & " (" & TempAmbU & ")"
TempLL = cCOMPASS.UnitConversion (CDbl(TempAmbLL),CInt(TempAmbU),CInt(TempAmbLunit),1)
'	cDebug.LogStatus "AmbTempLL = " & TempLL & " " & TempAmbUtxt & " (" & TempAmbU & ")"
TempHL = cCOMPASS.UnitConversion (CDbl(TempAmbHL),CInt(TempAmbU),CInt(TempAmbLunit),1)
'	cDebug.LogStatus "AmbTempHL = " & TempHL & " " & TempAmbUtxt & " (" & TempAmbU & ")"
If TempAmb < TempLL  Then
	msg = "Amb Temperature is too low: " & TempAmb & " " & TempAmbUtxt
	ambAlert msg,emailto,delay
ElseIf TempAmb > TempHL Then 
	msg = "Amb Temperature is too high: " & TempAmb & " " & TempAmbUtxt 
	ambAlert msg,emailto,delay
End If

RH_Amb = cCompass.cConfig.AmbRH.FinalOutput
RH_AmbUtxt = RH_AmbLtxt
'	cDebug.LogStatus "AmbRH = " & RH_Amb & RH_AmbUtxt
'	cDebug.LogStatus "AmbRH_HL = " & RH_AmbHL & RH_AmbLtxt
'	cDebug.LogStatus "AmbRH_LL = " & RH_AmbLL & RH_AmbLtxt
If RH_Amb < CDbl(RH_AmbLL)Then
	msg = "Amb RH is too low: " & RH_Amb & " " & RH_AmbUtxt
	ambAlert msg,emailto,delay
ElseIf RH_Amb > CDbl(RH_AmbHL) Then
	msg = "Amb RH is too high: " & RH_Amb & " " & RH_AmbUtxt
	ambAlert msg,emailto,delay
End If

End Function

Function ambAlert (msgAlert,toEmail,delay)
	msgAlert = msgAlert  & ". Waiting " & formatnumber (delay/60,1)& " minutes until checking again. "
'	cCOMPASS.StatusDisplay msgAlert
	cDebug.LogStatus msgAlert
	msgAlert = msgAlert & "Current Date/time is " & Date & ", " & time
	msgbox msgAlert, vbCritical, "Alert"
	subject = msgAlert
	SendEmail toEmail, subject, msgAlert
		cDebug.LogStatus "Sent email to: " & toEmail
	TimeDelayWstatus (delay)
End Function


The above macro calls the Global macro TimeDelayWstatus. TimeDelayWStatus should be added as a Global macro
'****************************************
'Delay for the amount of time in seconds
'specified by tD. Use the COMPASS object
'to allow for asynchronous operation
'****************************************
Sub TimeDelayWstatus(tD)
	tStart = Timer'Timer function returns the number of seconds since 12:00 AM
	Do 			
		cCOMPASS.TimeDelay 1
		tDiff = Time_Difference(tStart)
		remainT = CInt(tD - tDiff)
		cCOMPASS.StatusDisplay "Delay remaining..." & remainT
		If cCOMPASS.SystemAbort Then Exit Sub
		If remainT <= 0 Then Exit Do
	Loop Until False
End Sub

And this macro calls the Global macro TimeDelay. It is already in COMPASS as a Global macro but the text is here if you need to add it.
'****************************************
'Delay for the amount of time in seconds
'specified by tD. Use the COMPASS object
'to allow for asynchronous operation
'****************************************
Sub TimeDelay(tD)
cCOMPASS.TimeDelay tD
 End Sub

Same with Global macro Time_Difference
'///////////////////////////////////////////////////////////////
'Return the time difference from the current
'time and the time value passed T:
'
'In the event the diff is <0 then add the
'number of seconds in a day because midnight
'has passed since the start time.
'
'NOTE
'This function as an effective time resoluteion
'of 10 ms.
'
'///////////////////////////////////////////////////////////////
Public Function Time_Difference(t) 
Dim T_Diff 

T_Diff = Timer - t  
If T_Diff < -0.1 Then
    T_Diff = T_Diff + 86400
ElseIf T_Diff < 0 Then
    T_Diff = 0
End If
 
Time_Difference = T_Diff

End Function
